Configuration of a 2-18 GHz Synthesized Frequency Source Using the 8620C Sweep Oscillator



Designing and testing telecommunication equipment, radar, telemetry systems, and instrumentation often requires an ultra-stable microwave source. The additional requirement of automating these operations suggests a programmable frequency synthesizer.

This application note describes the configuration of a calculator-controlled 2-18 GHz synthesized frequency source using the Hewlett-Packard Interface Bus (HP-IB) as the common link between the instruments and the calculator. This source can then be easily integrated into an entire measurement system via the HP-IB or it can be expanded to provide 1 MHz to 18.6 GHz coverage with slight modification.

The key components of the source include the HP 8620C/86290A 2-18 GHz Sweep Oscillator, the HP 8660 UHF Synthesizer, the HP-IB, and a programmable calculator such as the HP 9820A, 9821A, or 9830A. For operation with the HP 8620A mainframe see Application Note 187-1.

APPLICATION NOTE 187-2

CONFIGURATION BLOCK DIAGRAM

In this application the microwave sweeper acquires the accuracy and stability of the UHF synthesizer through phase-locking. The basic technique involves generating harmonics of the synthesized reference and mixing with the sweeper signal. The resultant difference frequency is then compared with a 20 MHz standard which can either be the internal crystal in the synchronizer or, for more stability, the synthesizer's 10 MHz standard which would then be doubled and injected into the synchronizer.

Within the sweeper, the 2-18 GHz signal is generated by taking the fundamental, second, or third harmonic of a 2-6.2 GHz oscillator. See Figure 1. A side benefit of this multiplier approach is the availability of the 2-6.2 GHz "AUX OUT" signal at a rear panel connector. Although the sweeper signal in the phase-lock loop could be sampled at the main RF output, this would require a broadband coupling device and harmonic mixer capable of producing acceptable harmonic content from 2-18 GHz. It is preferable, therefore, to use the "AUX OUT" since phase-locking this signal will also lock the output.

The reference and "AUX OUT" signal enter the harmonic mixer through a Type N Tee. Since the harmonic amplitude in the mixer decreases and becomes less predictable as the harmonic number increases, it makes sense to use the highest reference frequency (lowest harmonic number) which is available. Therefore, the frequency of the reference is determined by the equation

$$F_{REF} = \frac{\frac{F_0}{N_1} - 20\text{ MHz}}{N_2}$$

where F_0 is the desired output frequency; N_1 is the multiplier of the 86290A's fundamental oscillator (i.e., either 1, 2, or 3) and N_2 is the lowest integer which will result in the $F_{REF} \leq 1300$ MHz. N_2 will be between 2 and 5. Once F_0 is input to the calculator, it will determine F_{REF} and through the HP-IB instruct the 8660 to output that frequency. It will also tune the 86290A to within 20 MHz of F_0. The phase-lock loop will then fine tune the 86290A producing a synthesized F_0.

SUGGESTED COMPONENTS

Sweep Oscillator Mainframe with HP-IB Programming Option	HP8620C Opt 011
2-18 GHz Sweep Oscillator Plug-in	HP86290A
Synthesized Signal Generator with HP-IB Option	HP8660A Opt 005
Auxiliary Section	HP86631B
1300 MHz RF Section	HP86602B
Frequency Extension Module	HP11661B
Harmonic Mixer	HP8709A-K18
Type N Tee	HP1250-0846
Synchronizer	HP8709A



Figure 1 2-18 GHz Synthesizer Configuration

Programmable Calculator with appropriate ROMS and Accessories	HP9820A, 9821A, or 9830A
Calculator Printer (9830A only)	HP9866A
HP-IB Calculator Interface	HP59405A
Math ROM Block (9820A, 9821A only)	HP11221A
ASCII Interface Cables	HP10631A, B, or C

OPERATION

Connect equipment as shown in Figure 1.

8620C/86290A

Rear panel FM/NORM/PL switch must be in PL (phase-lock)

Ensure that Listen Address = "&"

Other plug-in controls are determined by power level required and leveling mode used.

8660

Ensure that Listen Address = "3"

8709A

Set MOD SENS switch to 6.0 MHz/VOLT

9820A, 9821A, or 9830A

Install the 59405A HP-IB calculator interface and (for the 9820A or 9821A) the 11221A math ROM block.

Ensure that the 9866A Calculator Printer is attached when using the 9830A.

Key into the calculator the program appropriate to the calculator being used (Figure 4 or 5). The program will ask the user whether a "CW" frequency or digitally "STEPPED" frequencies are desired. After keying in this and other inputs, press "RUN" (9820A/21A) or "EXECUTE" (9830A). If "CW" is chosen, it will ask the user to input the "NEXT FREQ (MHz)." This frequency can be anywhere between 2000 and 18000 MHz. Once this frequency is entered, the calculator will set the sweeper and synthesizer to the proper frequencies. Then the phase-lock loop takes over and removes any error in the output frequency. This frequency remains until another frequency is input from the keyboard.

If "STEP" is chosen the user has the alternative of stepping through the desired range a "SINGLE" time or in a "REPETITIVE" mode. The user then is asked to input "F-START (MHz)" and "F-STOP (MHz)" with the same limits as "CW" and "STEP-SIZE (MHz)" which can be as small or large as desired. The calculator begins with F-START and steps through the frequency range at the rate of approximately one frequency change per second. To exit from the program push "STOP."

TYPICAL PERFORMANCE CHARACTERISTICS

The attached programs allow 2.000 to 18.000 GHz operation. Output Power level is >+5 dBm. Frequency accuracy, stability, and purity are almost entirely dependent on the reference oscillator used. For example, using the 8660 Opt 001 Synthesized Signal Generator as reference and as injection-lock source for the 8709A, typical accuracy and stability can be ± 54 Hz/day (±3 parts in 10^9) at 18 GHz. Resolution and step-size are typically 15 Hz. Single sideband phase noise is approximately equal to the Reference Oscillator phase noise plus 20 log ($N_1 x N_2$) plus approximately 10 dB between 2 kHz and 15 kHz offset from the carrier and 5 dB elsewhere, due to the phase-locked loop. See Figure 2. For example, at 25 kHz from the 3 GHz carrier the SSB phase noise is 90 dB down in a 1 Hz BW. Settling time to within 1 kHz is approximately 100 msec.



Figure 2 Actual Phase-Noise Measurement. Two phase-locked systems in phase quadrature made in a 1 kHz BW and scaled to a 1 Hz BW.

SPECIAL CONSIDERATIONS

Special sweeper characteristics are extremely important for successful operation:

A) Frequency accuracy plus programming resolution must be within the capture range of the synchronizer (~35 MHz).

B) An auxiliary output which is related to RF output frequency, but which does not exceed the range of the harmonic mixer (~6th harmonic) is required to ensure adequate harmonic content.

C) There must be reasonable overlap at band switchovers to allow sufficient analog frequency control.

D) Harmonic-related signals from the sweep oscillator must be low enough to prevent falselocking.

EXPANDED CAPABILITIES

In applications requiring a frequency or phase modulated source this can be accomplished by modulating the reference oscillator. This involves inserting the proper modulation section in the 8660 mainframe. The deviation of the microwave output will be that of the reference multiplied by N_1 x N_2.

Since the pin modulator in the 86290A is outside the phase-locked loop (after the AUX OUT coupler) it has relatively little effect on the frequency. In most AM applications, therefore, as long as the 86290A is in the leveled mode, no external modulator is necessary. This includes internal 1 kHz square wave modulation and the 27.8 kHz square wave modulator drive signal from the 8755 Swept Amplitude Analyzer. It should be noted, however, that some incidental FM may be produced at high AM rates (>100 kHz).

If operation to 18.6 GHz is desired, it can be obtained by special programming of the 8620C. The sweeper requires >10V to produce >18 GHz. Since the 8620C allows only four digits of input (i.e., V.XXX) there is a special symbol which denotes V = 10. The ASCII character for that symbol is ":". For example, a frequency of 18.3 GHz requires 10.500V; the required ASCII string would be ":500". Instructions for modifying the programs are included at the end of each one.

Synthesized operation from 1 MHz to 18.6 GHz is possible using the 86603A 2.6 GHz plug-in for the 8660 instead of the 86602B. The output is then switched using two 8761A/B coax switches controlled by a 59306A Relay Actuator. A 12V or 24V power supply is also required. The software must be modified to set the proper 1 MHz to 2 GHz frequencies and to include control of the Relay Actuator.

The utility of the source is greatly enhanced by the HP-IB which controls it. With the HP-IB a total automatic measurement system can be configured. Up to 12 additional instruments can be controlled. For example, the 436A Power meter could be added to the HP-IB to control output power of the source to approximately the accuracy and resolution of the power meter. Or a programmable attenuator could be added to control power over a 110 dB range.

Of course, since the calculator is part of the system, it can be used for error correction, manipulation of data, and printout of required information. It can also be used as a diagnostic aid.



Figure 3 Program Flow Diagram

9830A PROGRAM LISTING

```
10 REM          2-18 GHZ FREQUENCY SYNTHESIZER PROGRAM
20 REM         SEPTEMBER 19, 1975
30 REM           U=9830;3=8660;&=8620C
40 DIM DI[8]
50 REM   D(1,2,3)=LOW LIMITS OF BANDS 1,2,3
60 REM   D(4,5,6)=WIDTH OF BANDS 1,2,3
70 REM   D(7,8)=BAND SWITCH POINTS
80 D[1]=2000
90 D[2]=D[6]=6000
100 D[3]=12000
110 D[4]=4200
120 D[5]=6400
130 D[7]=6100
140 D[8]=12200
150 REM   U,L=UPPER,LOWER LIMITS OF SWEEPER
160 U=18000
170 L=2000
180 DISP "STEP:1 OR CW:2";
190 INPUT M
200 GOTO M OF 210,340
210 DISP "SINGLE:1 OR REPET.:2";
220 INPUT N
230 DISP "START FREQ(MHZ)";
240 INPUT F1
250 IF F1<L OR F1>U THEN 840
260 DISP "STOP FREQ(MHZ)";
270 INPUT F2
280 IF F2<L OR F2>U THEN 840
290 DISP "STEP-SIZE(MHZ)";
300 INPUT F3
310 A=F1
320 GOTO 370
330 PRINT "FREQ = "A" MHZ"
340 DISP "NEXT FREQ(MHZ)";
350 INPUT A
360 IF A<L OR A>U THEN 840
370 GOSUB 580
380 GOSUB 730
390 IF M=2 THEN 330
400 WAIT 1000
410 IF A >= F2 THEN 480
420 A=A+F3
430 REM  NEXT IF STMT ALLOWS COVERAGE OF F2
440 REM  EVEN IF (F2-F1)/F3 IS NOT AN INTEGER
450 IF A <= F2 THEN 370
460 A=F2
470 GOTO 370
480 IF N=3 THEN 540
490 PRINT "START FREQ="F1"MHZ"
500 PRINT "STOP FREQ ="A"MHZ"
510 PRINT "STEP-SIZE ="F3"MHZ"
520 IF N=1 THEN 230
530 N=3
540 PRINT "REPEATING"
550 DISP " "
560 A=F1
570 GOTO 370
580 REM
590 REM    SWEEPER TUNE SUBROUTINE
600 REM
610 R6=(A>D[7])+(A>D[8])
620 REM   R6=0,1,2 FOR BANDS 1,2,3
630 C=(A-D[1+R6])*10/D[4+R6]
640 REM   C=(% OF BAND)/10
650 CMD "?U&"
660 IF C<9.999 THEN 700
670 REM   FOR OPERATION ABOVE 18GHZ SEE BELOW
680 C=9.999
690 FORMAT "M1B",F1000.0,"V",F1000.3,"E"
700 OUTPUT (13,690)R6+1,C
710 RETURN
720 REM
730 REM   SYNTHESIZER TUNE SUBROUTINE
740 REM
750 R3=A/(R6+1)-20
760 R3=R3/(INT(R3/1300)+1)
770 R3=INT(R3*1E+06)/1E+09
780 GOSUB 890
790 CMD "?U3"
800 FORMAT F1000.0,"(",F1000.0,"C"
810 OUTPUT (13,800)Z,300
820 RETURN
830 REM
840 REM    OUT OF RANGE SUBROUTINE
850 REM
860 PRINT "OUT OF RANGE!!!!"
870 WAIT 5000
880 GOTO M OF 210,340
890 REM
900 REM    8660 INVERSION SUBROUTINE
910 REM
920 W=1
930 Y=ABSR3
940 Z=0
950 Z=Z+W*INTY
960 Y=10*(Y-INTY)
970 W=10*W
980 IF Y#0 THEN 950
990 RETURN
1000 END

    FOR OPERATION TO 18.6 GHZ
    --------------------------
    CHANGE OR ADD LINES BELOW
160 U=18600
660 IF C<9.9995 THEN 700
670 C=C-10
672 IF C<0.999 THEN 680
673 C=0.999
675 FORMAT "M1B3V:",F1000.3,"E"
680 OUTPUT (13,675)ABSC
685 RETURN
```

Figure 4